/*
Jonathan Chu	
Proposal title: Race, Religion, and American Support for Humanitarian Intervention


HYPOTHESES

Stated-Hyp1: Citizens are more supportive of saving the lives of those who belong to their
in-group compared to outgroup (p. 2)

	Test-Hyp1: Respondents will be more supportive of saving the lives of those of the same religion compared to those of a different religion.
		NOTE: only testable for Christian and Muslim respondents
			
	Test-Hyp2: Respondents will be more supportive of saving the lives of those of the same race compared to those of a different race.
		NOTE: only testable for black and white respondents
	
Stated-Hyp2: Cosmopolitans (those who see themselves as citizens of the world) exhibit lower levels of racial/religious bias than non-Cosmopolitans (p. 3)

	Test-Hyp3: Respondents will be more supportive of saving the lives of those of the same religion compared to those of a different religion, but this effect will be lower for those with cosmopolitan vs. non-cosmopolitan values.
		NOTE: only testable for Christian and Muslim respondents
			
	Test-Hyp4: Respondents will be more supportive of saving the lives of those of the same race compared to those of a different race, but this effect will be lower for those with cosmopolitan vs. non-cosmopolitan values.
		NOTE: only testable for black and white respondents
	

Stated-Hyp3: Americans are more supportive of humanitarian intervention on behalf of Black Africans.
Furthermore, this effect should be moderated by exposure to media, and mediated by people’s feelings of paternalism (p. 3)

	Test-Hyp5: Respondents will be more supportive of intervention to help Black Africans

	Test-Hyp6: Support for intervention to help Black Africans will be moderated by exposure to media

	Test-Hyp7: Support for intervention to help Black Africans will be mediated by feelings of paternalism

*/


use "Chu958.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATION

* race and religion scenario
	tab SCENARIO
	recode SCENARIO (1/2=2) (3/4=1) (5=0), gen(treat_religion)
	lab def religion 2 "2 Muslim" 1 "1 Christian" 0 "0 not stated"
	lab val treat_religion religion
	tab treat_religion

	recode SCENARIO (1=1) (2=2) (3=1) (4=2) (5=0), gen(treat_race)
	lab def race 1 "1 white" 2 "2 black" 0 "0 not stated"
	lab val treat_race race
	tab treat_race
	
	* to test stated hyp 3
	recode treat_race (1=0) (0=0) (2=1), gen(treat_black)
	tab treat_black

* R's race in-group vs. out-group
	gen race_ingroup = . 
	replace race_ingroup=1 if RACETHNICITY==1 &treat_race==1
	replace race_ingroup=1 if RACETHNICITY==2 &treat_race==2
	replace race_ingroup=0 if RACETHNICITY==1 &treat_race==2
	replace race_ingroup=0 if RACETHNICITY==2 &treat_race==1
	tab race_ingroup, mis
	// missing cases are respondents who are not black or white. 
	
* R's religion ingroup vs. out-group
	gen relig_ingroup = . 
	replace relig_ingroup=1 if Q7==1 &treat_religion==1
	replace relig_ingroup=1 if Q7==2 &treat_religion==2
	replace relig_ingroup=0 if Q7==1 &treat_religion==2
	replace relig_ingroup=0 if Q7==2 &treat_religion==1
	tab relig_ingroup, mis	
	// missing cases are respondents who are not christian or muslim
	
* R is cosmopolitan
	/* two vars are reverse coded; correcting that
	also changing missing value to .
	*/
	lab def cosmo 1 "strongly disagree" 5 "strongly agree"
	
	foreach var in Q1A Q1C {
	recode `var' (5=1) (4=2) (3=3) (2=4) (1=5) (*=.)
	lab val `var' cosmo
	tab `var', mis
	}
	
	foreach var in Q1B Q1D {
	replace `var'=. if `var'>5
	tab `var', mis
	}

	// NOTE: Q1B and Q1D are correctly coded: higher values indicate higher cosmopolitanism
	
	// creating summated scale
	gen cosmopolitan = Q1A + Q1B + Q1C + Q1D
	sum cosmopolitan
	
* R's media exposure
	gen follownews= Q4
	replace follownews=. if follownews>5
	lab def media 1 "never" 5 "everyday"
	lab val follownews media
	tab follownews

* R's paternalism
	/* [The civilians have the ability to help themselves and do not need outside support.] Do you agree or disagree with the following statements about the situation you just read?
	
	Higher values indicate higher levels of paternalism.
	*/
	gen paternalism = Q3B
	replace paternalism = . if paternalism>5
	tab paternalism
	

* OUTCOME MEASURES

* support for US military action
	recode Q2 (7=1) (6=2) (5=3) (4=4) (3=5) (2=6) (1=7) (*=.), gen(support)
	lab def support 1 "oppose a great deal" 7 "support a great deal
	lab val support support
	tab support, mis
	
********************************************************************************

* ANALYSIS
	
*Test-Hyp1: Respondents will be more supportive of saving the lives of those of the same religion compared to those of a different religion.
	reg support i.relig_ingroup
	// do not reject. p= 0.000 
	tess 1.relig_ingroup +, init(Chu958)
	
*Test-Hyp2: Respondents will be more supportive of saving the lives of those of the same race compared to those of a different race.
	reg support i.race_ingroup
	// do not reject. p= 0.033 
	tess 1.race_ingroup +
	
*Test-Hyp3: Respondents will be more supportive of saving the lives of those of the same religion compared to those of a different religion, but this effect will be lower for those with cosmopolitan vs. non-cosmopolitan values.
	reg support i.relig_ingroup##c.cosmopolitan
	// do not reject. p= 0.000
	tess 1.relig_ingroup#c.cosmopolitan -			
			
*Test-Hyp4: Respondents will be more supportive of saving the lives of those of the same race compared to those of a different race, but this effect will be lower for those with cosmopolitan vs. non-cosmopolitan values.
	
	reg support i.race_ingroup##c.cosmopolitan
	// do not reject. p= 0.028
	tess 1.race_ingroup#c.cosmopolitan -	
	
*Test-Hyp5: Respondents will be more supportive of intervention to help Black Africans

	reg support i.treat_black
	tess 1.treat_black +
	// reject. p=0.853

*Test-Hyp6: Support for intervention to help Black Africans will be moderated by exposure to media

	reg support i.treat_black##c.follownews
	tess 1.treat_black#c.follownews	
	// reject. p=0.830

*Test-Hyp7: Support for intervention to help Black Africans will be mediated by feelings of paternalism

	reg support treat_black
	est sto m1

	reg support treat_black paternalism
	est sto m2
	suest m1 m2

	test [m1_mean]treat_black=[m2_mean]treat_black
	
	tess, mediation
	// reject. p=0.8217

